
■■■■ 適応ハウリング適応ハウリング適応ハウリング適応ハウリング・・・・キャンセラのバイアス蓄積問題のシミュレーションキャンセラのバイアス蓄積問題のシミュレーションキャンセラのバイアス蓄積問題のシミュレーションキャンセラのバイアス蓄積問題のシミュレーション 2025.6.30

（有）ケプストラム

適応ハウリング・キャンセラの適応フィルタは、教科書に書いてあるような理想的な収束特性を有しては

おらず、バイアス蓄積という厄介な問題が存在することをご存知ない方が多いようなので、簡単な説明資料を

作成しました． バイアス蓄積の問題に関連した簡単なシミュレーション結果をご紹介します．

シミュレーションに用いたプログラミング言語は Scilab で、ソースはこのPDFファイルに添付してあります．

適応ハウリング・キャンセラの実用化にあたって、適応フィルタのバイアス蓄積は解決すべき問題の一つ

です． バイアス蓄積によって発生する現象の一つが適応フィルタの低域の利得上昇で、際限なく利得が増える

ために動作が不安定になってしまいます．（図１） 場合によっては高域（fs/2端）の利得が上昇する場合も

あります．

図１ バイアス蓄積によって生じる適応フィルタの低域の利得上昇例

高域（fs/2）端の利得上昇が生じる場合もある

長い時間を必要としましたが、当社ではバイアス蓄積問題をなんとか解決することが出来ました．

複数の要因が複雑にからみあってバイアス蓄積が生じていると考えられますが、最終的に各種動作パラメータの

最適化、dither信号付加等の対策を適用しました．

きちんとした理論的な解析は出来なかったのですが、適応フィルタのバイアス蓄積の要因として以下のような

ものが考えられます．

・音響系の伝達特性（インパルス・レスポンス）の変化・揺らぎ

・周期性成分を有する暗騒音

・各種時変補助処理によって発生する歪

・decorrelatorの性能の制約（周期音に対して十分な無相関化処理が出来ない）

・decorrelatorの周波数シフト／変調処理による、歪／バイアスの処理帯域全体へのスペクトラム拡散

・入力信号の帯域制限による処理帯域両端での適応フィルタの収束特性の低下 etc

最後の帯域制限に関しては、簡単なシミュレーションで何が問題なのかを確認できるので、シミュレーション

結果の一例を以下にご紹介します．

Adaptive Howling Canceller Bias Problem Simulation

(Adaptive Acoustic Feedback Canceller Bias Problem Simulation)

https://www.cepstrum.co.jp/

適応ハウリング・キャンセラの原理的な基本構成にBPF（バンドパス・フィルタ）を追加したものを図２に

示します． 室内の暗騒音は一般的に低域上昇特性を有していて、低域の周期性成分を含む空調騒音・自動車

騒音等は適応フィルタの収束特性に悪影響を与えるので、実用レベルのハウリング・キャンセラではBPFにより

音声帯域外の成分を除去する必要があります． 動作条件によっては高域側を帯域制限せず、BPFの代わりに

HPF（ハイパス・フィルタ）を用いても問題ありません．

図２のようにBPFを追加すると、システム同定の構成の適応システム（System Identification Adaptive

System）の入力信号も帯域制限されることになります． 入力信号の帯域制限により、適応フィルタの収束

特性が変化することは簡単なシミュレーションで確認できます．

図２ BPF（バンドパス・フィルタ）を追加した適応ハウリング・キャンセラの基本構成

BPF追加の影響確認のシミュレーションは、図３の基本的なシステム同定の構成の適応フィルタでおこない

ます． 同定対象のシステムをバンドパス特性（BPF）として、入力信号の帯域制限の有無による収束特性の

違いを比較するだけです． 適応アルゴリズムには時間領域のLMSアルゴリズムを用いています．

図３ 同定対象がバンドパス特性（BPF）を有するシステム同定の構成の適応システム

入力信号の帯域制限の有無による収束特性の違いを比較する

サンプリング周波数は8kHzとしてシミュレーションをおこなっています． 同定対象のBPFの通過帯域は

250Hz～3900Hzです．（図４）

適応フィルタの入力信号には白色雑音または帯域制限した白色雑音を用います． シミュレーションに用いた

帯域幅250Hz～3900Hzに帯域制限した白色雑音の周波数特性を図５に示します．

図４ 同定対象のBPFの周波数特性 図５ 帯域幅250Hz～3900Hzに帯域制限した

白色雑音の周波数特性

シミュレーション結果（収束後の適応フィルタの周波数特性）を図６に示します．

白色雑音を用いた場合の収束特性は極めて良好です．（図６左）

一方、帯域制限した白色雑音を用いると、DC端・fs/2端で十分に利得が下がっていないことが分かります．

（図６右） このような収束特性の劣化と、他の要因により発生する歪／バイアス、decorrelatorの周波数

シフト／変調処理による歪／バイアスの帯域全体への拡散等が絡み合って適応フィルタのバイアス蓄積の問題が

生じていると考えられます．

図６ シミュレーション結果（収束後の適応フィルタの周波数特性）

左：入力信号は白色雑音 右：入力信号は帯域制限した白色雑音

同定対象のPlantが帯域制限特性を持ち、入力信号も帯域制限されていると

適応フィルタのDC端・fs/2端の収束特性が劣化する

バイアス蓄積と類似の問題は逆フィルタの構成の適応システム（図７）でも発生することが知られています．

図７においてH(ω)／Plantの周波数特性が鋭どいヌル（null）を持つと、逆特性を有する適応フィルタADPの

利得が極端に大きくなり、収束特性が不安定になる場合があります． この問題を回避する簡単な方法は、

Plantの利得の低い周波数帯域にエネルギーを有するnoiseを付加することです．

図７ 逆フィルタの構成の適応システム（逆フィルタ設計）

H(ω)／Plantの周波数特性に鋭どいヌル（null）があると、

逆特性の適応フィルタADPの利得が極端に大きくなるために収束特性が不安定になる

ヌル（null）の周波数帯域にエネルギーを有するnoise付加により安定性の問題に対処できる

同様に当社の適応ハウリング・キャンセラでも帯域制限されたシステム／入力信号によって発生するバイアス

蓄積問題に対処するために、微小量のdither信号を付加していますが、その効果はdither付加単独で得られる

ものではありません．（他のバイアス蓄積低減対策も必要です）

----------------------------------------------------------------------------------------------------

添付ファイル

sim0001.sce Scilabシミュレーション・プログラム

sim0002.sce Scilabシミュレーション・プログラム

bp_noise.wav 帯域制限白色雑音wavファイル（sim0002.sceの実行に必要です）




// https://www.cepstrum.co.jp/

//         

//                        +---+ d             

//                   +--->| h +---+     

//                   |    +---+   |      

//                   |          + v  e    

// white noise  x ---+      ^    (+)---+--->

//                   |      |   - ^    |   

//                   |    +---+   |    |

//                   +--->| w +---+    |

//                        +---+ y      |

//                          |          |

//                          +----------+



clear;                   // clear all variables



rand('normal');



DATLEN=200000;           // data length 

FS    =8000;             // sampling frequency [Hz]

FIRLEN=1024;             // adaptive filter legnth

GAIN  =4.0;              // h() gain

MU    =0.0003;           // step size parameter of adaptive filter



indata=rand(1:DATLEN);                  // generate white noise

indata=indata/max(abs(indata));         // scaling 

outdata=zeros(1:DATLEN);



xbuf=zeros(1:FIRLEN);

w=zeros(1:FIRLEN);



// BPF coefficient, Butterworh 250Hz/3900Hz (fs=8kHz)

A = 8.233127742944482685D-01;

B1=-1.652569066320155322D-01;

B2= 1.603848922725006343D+00;

B3= 1.115318355518675197D-01;

B4=-6.779093254568269167D-01;



// generate impulse response h()

rand('seed', 3141);

h=[zeros(1:5), rand(1:FIRLEN-5)];

h=h.*(0.975^(1:FIRLEN)); 

h=h/max(abs(fft(h, -1)));

h=GAIN*h;



// apply BPF to h(), Butterworth 250Hz/3900Hz

zx=0; zzx=0; zzzx=0; zzzzx=0; 

zy=0; zzy=0; zzzy=0; zzzzy=0;

for i=1:FIRLEN

  x=h(i);

  y=A*(x-2.0*zzx+zzzzx)+B1*zy+B2*zzy+B3*zzzy+B4*zzzzy;

  zzzzx=zzzx; zzzx=zzx; zzx=zx; zx=x; 

  zzzzy=zzzy; zzzy=zzy; zzy=zy; zy=y;

  h(i)=y;

end



scf(0);

clf;

plot(h);        // plot h()



// plot h() frequency response

scf(1);

clf;

s=20.0*log10(abs(fft(h, -1))+1e-99);

s=s(1:FIRLEN/2);

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=1);

xlabel "frequency [Hz]"

ylabel "[dB]"



for i=1:DATLEN 

  x=indata(i);



  xbuf=[x, xbuf(1:FIRLEN-1)];

  d=h*xbuf';

  y=w*xbuf';

  e=d-y;

  w=w+2.0*MU*e*xbuf;       // LMS algorithm



  outdata(i)=e;

end



scf(2);

clf;

plot(outdata);          // plot e



// plot w() frequency response

s=20.0*log10(abs(fft(w, -1))+1e-99);

s=s(1:FIRLEN/2);

scf(3);

clf;

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=2);

xlabel "frequency [Hz]"

ylabel "[dB]"




// https://www.cepstrum.co.jp/

//         

//                         +---+ d             

//                    +--->| h +---+     

//                    |    +---+   |      

//                    |          + v  e    

// band limited  x ---+      ^    (+)---+--->

// white noise        |      |   - ^    |   

//                    |    +---+   |    |

//                    +--->| w +---+    |

//                         +---+ y      |

//                           |          |

//                           +----------+



clear;                   // clear all variables



rand('normal');



DATLEN=200000;           // data file length 

FS    =8000;             // sampling frequency [Hz]

FIRLEN=1024;             // adaptive filter legnth

GAIN  =4.0;              // h() gain

MU    =0.0001;           // step size parameter of adaptive filter



indata=loadwave('bp_noise.wav');     // read band limited white noise wav file

outdata=zeros(1:DATLEN);



xbuf=zeros(1:FIRLEN);

w=zeros(1:FIRLEN);



// BPF Butterworh, 250Hz/3900Hz (fs=8kHz)

A = 8.233127742944482685D-01;

B1=-1.652569066320155322D-01;

B2= 1.603848922725006343D+00;

B3= 1.115318355518675197D-01;

B4=-6.779093254568269167D-01;



// generate impulse response of acoustice system (h)

rand('seed', 3141);

h=[zeros(1:5), rand(1:FIRLEN-5)];

h=h.*(0.975^(1:FIRLEN)); 

h=h/max(abs(fft(h, -1)));

h=GAIN*h;



// apply BPF to h(), Butterworth 250Hz/3900Hz

zx=0; zzx=0; zzzx=0; zzzzx=0; 

zy=0; zzy=0; zzzy=0; zzzzy=0;

for i=1:FIRLEN

  x=h(i);

  y=A*(x-2.0*zzx+zzzzx)+B1*zy+B2*zzy+B3*zzzy+B4*zzzzy;

  zzzzx=zzzx; zzzx=zzx; zzx=zx; zx=x; 

  zzzzy=zzzy; zzzy=zzy; zzy=zy; zy=y;

  h(i)=y;

end



scf(0);

clf;

plot(h);        // plot h()



// plot h() frequency response

scf(1);

clf;

s=20.0*log10(abs(fft(h, -1))+1e-99);

s=s(1:FIRLEN/2);

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=1);

xlabel "frequency [Hz]"

ylabel "[dB]"



for i=1:DATLEN 

  x=indata(i);



  xbuf=[x, xbuf(1:FIRLEN-1)];

  d=h*xbuf';

  y=w*xbuf';

  e=d-y;

  w=w+2.0*MU*e*xbuf;       // LMS algorithm



  outdata(i)=e;

end



scf(2);

clf;

plot(outdata);          // plot e



// plot w() frequency response

s=20.0*log10(abs(fft(w, -1))+1e-99);

s=s(1:FIRLEN/2);

scf(3);

clf;

plot2d((FS/2.0)*(0:(FIRLEN/2-1))/(FIRLEN/2), s, axesflag=1, rect=[0, -60, FS/2, 20], style=5);

xlabel "frequency [Hz]"

ylabel "[dB]"



